Linear One-dimensional Site Response Analysis in the presence of stiffness-less free surface for certain power-law heterogeneities

Visualize the conditions

As discussed in the text, Section 3, assuming power-law behavior $O(z^k)$ in the displacement field and stiffness scaling $O(z^n)$ in the stiffness.

Traction-free surface condition:

```
In[1]:= Tractions = n + k - 1 > 0;
```

Finitude-of-deformation-energy condition:

```
In[2]:= Kinetic = n + 1 + 2 (k - 1) > 0;
```

Finitude-of-kinetic-energy condition:

```
In[3]:= Deformation = 1 + 2 k > 0;
```

In[4]:=
```
figCond = RegionPlot[{Tractions, Deformation, Kinetic, Tractions && Deformation && Kinetic},
  {k, -1, 1}, {n, 0, 4}, PlotLegends → "Expressions",
  FrameLabel → {k, n}, RotateLabel → False, LabelStyle → (FontSize → 15)]
```

Out[4]=



The previous image is Figure 2 in the text.

Solution of the governing equation

General solution for any **n**

The equation (6) corresponds to a Bessel equation, and so do the solutions:

In[5]:=
$$\texttt{Factor}\left[\texttt{Refine}\left[\texttt{DSolve}\left[u''[\xi] + \frac{n}{\xi}u'[\xi] + \frac{r^2}{\xi^n}u[\xi] == 0, u[\xi], \xi\right], \{n > 0, \xi > 0\}\right]\right]$$

Out[5]=
$$\left\{\left\{u[\xi] \to \left(-1+\frac{2}{n}\right)^{\frac{1}{-1+\frac{2}{n}}-\frac{1}{\left(-1+\frac{2}{n}\right)n}} n^{\frac{1}{-1+\frac{2}{n}}-\frac{1}{\left(-1+\frac{2}{n}\right)n}} r^{-\frac{1}{-1+\frac{2}{n}}+\frac{1}{\left(-1+\frac{2}{n}\right)n}} \xi^{\left(-\frac{1}{2}+\frac{1}{n}\right)\left(-\frac{1}{-1+\frac{2}{n}}+\frac{1}{\left(-1+\frac{2}{n}\right)n}\right)n} \left(\texttt{BesselJ}\left[\frac{1-n}{-2+n}, \frac{2\,r\,\xi^{\frac{1}{2}\left(-1+\frac{2}{n}\right)n}}{\left(-1+\frac{2}{n}\right)n}\right] C[1]\,\texttt{Gamma}\left[\frac{1}{2-n}\right] + \texttt{BesselJ}\left[\frac{-1+n}{-2+n}, \frac{2\,r\,\xi^{\frac{1}{2}\left(-1+\frac{2}{n}\right)n}}{\left(-1+\frac{2}{n}\right)n}\right] C[2]\,\texttt{Gamma}\left[-\frac{3}{-2+n}+\frac{2n}{-2+n}\right]\right)\right\}\right\}$$

Simplify for the case **0<n<2**

Simplify the multiplying factor scaling the whole solution

In[6]:= $$\text{Factor}\left[\text{FullSimplify}\left[\left(-1+\frac{2}{n}\right)^{\frac{1}{-1+\frac{2}{n}}-\frac{1}{\left(-1+\frac{2}{n}\right)n}} n^{\frac{1}{-1+\frac{2}{n}}-\frac{1}{\left(-1+\frac{2}{n}\right)n}} r^{-\frac{1}{-1+\frac{2}{n}}+\frac{1}{\left(-1+\frac{2}{n}\right)n}}, \{r>0, 0<n<2\}\right]\right]$$

Out[6]= $$-\frac{\left(\frac{2-n}{r}\right)^{\frac{1}{2-n}} r}{-2+n}$$

Simplify the exponent that appears multiplying the Bessel functions:

In[7]:= $$\text{Factor}\left[\text{FullSimplify}\left[\xi^{\left(-\frac{1}{2}+\frac{1}{n}\right)\left(-\frac{1}{-1+\frac{2}{n}}+\frac{1}{\left(-1+\frac{2}{n}\right)n}\right)n}, \{\xi>0, n>0\}\right]\right]$$

Out[7]= $$\xi^{\frac{1}{2}-\frac{n}{2}}$$

Simplify the exponent that appears within the Bessel functions:

In[8]:= $$\text{Factor}\left[\text{FullSimplify}\left[\xi^{\frac{1}{2}\left(-1+\frac{2}{n}\right)n}, \{\xi>0, n>0\}\right]\right]$$

Out[8]= $$\xi^{1-\frac{n}{2}}$$

Thus the general solution, eq.(7), in this case:

In[9]:= $$\text{gensol}[\xi\_, n\_, A\_, B\_] = \left(\frac{r}{2-n}\right)^{\frac{1-n}{2-n}} * \xi^{\frac{1-n}{2}} * \left(A * \text{BesselJ}\left[-\frac{1-n}{2-n}, \frac{2r}{2-n}\xi^{1-\frac{n}{2}}\right]\text{Gamma}\left[\frac{1}{2-n}\right] + B * \text{BesselJ}\left[\frac{1-n}{2-n}, \frac{2r}{2-n}\xi^{1-\frac{n}{2}}\right]\text{Gamma}\left[\frac{2n-3}{n-2}\right]\right);$$

See that the Gamma functions that appear in the constants have not been absorbed in the coefficients.

Asymptotic behavior of the exact solution around ξ=0

Using an auxiliary change of variables $x = \xi^{1-n/2}$:

In[10]:= $$\text{Simplify}\left[\xi^{\frac{1}{2}-\frac{n}{2}} \,/.\, \xi \to x^{\frac{1}{1-n/2}}, \{0<n<2, x>0\}\right]$$

Out[10]= $$x^{\frac{-1+n}{-2+n}}$$

<u>First portion of the solution</u>

In[11]:= $$\text{Simplify}\left[x^{\frac{1-n}{2-n}} * \text{Refine}\left[\text{Series}\left[\text{BesselJ}\left[-\frac{1-n}{2-n}, \frac{2r}{2-n}x\right], \{x, 0, 2\}\right], \{0<n<2, r>0\}\right]\right]$$

Out[11]= $$\frac{\left(\frac{r}{2-n}\right)^{\frac{1-n}{-2+n}}}{\text{Gamma}\left[\frac{1}{2-n}\right]} + \frac{r^2\left(\frac{r}{2-n}\right)^{\frac{1-n}{-2+n}}x^2}{(-2+n)\,\text{Gamma}\left[\frac{1}{2-n}\right]} + O[x]^3$$

Tends to a constant at the free-surface. See that the second term amounts to ξ^{2-n}, thus the strain (derivative) goes as ξ^{1-n}, hence it can be singular.

Second portion of the solution

In[12]:= $\text{Simplify}\left[x^{\frac{1-n}{2-n}} * \text{Refine}\left[\text{Series}\left[\text{BesselJ}\left[\frac{1-n}{2-n}, \frac{2\,r}{2-n}\,x\right], \{x, 0, 1\}\right], \{0 < n < 2, r > 0\}\right]\right]$

Out[12]= $x^{\frac{2\,(-1+n)}{-2+n}} \left(\frac{\left(\frac{r}{2-n}\right)^{\frac{-1+n}{-2+n}}}{\text{Gamma}\left[\frac{-3+2\,n}{-2+n}\right]} + O[x]^2 \right)$

The exponent of x is just twice the substitution we obtained before, hence goes as ξ^{1-n}, hence is singular if n>1.

This solution would not satisfy either the traction free condition or the deformation energy finiteness, thus will always be discharge

The solution after applying boundary conditions

Actual displacement field, eq.(10):

In[13]:= $u[\xi\_, r\_, n\_] = \xi^{\frac{1-n}{2}} \frac{\text{BesselJ}\left[-\frac{1-n}{2-n}, \frac{2\,r}{2-n}\,\xi^{1-\frac{n}{2}}\right]}{\text{BesselJ}\left[-\frac{1-n}{2-n}, \frac{2\,r}{2-n}\right]};$

Comparison of displacement fields:

In[14]:=
```
DE = 0.1;
RR = π/2;
Rotate[Plot[{Abs[u[ξ, RR/Sqrt[1 + ⅈ * DE], 4/3]], Abs[u[ξ, RR/Sqrt[1 + ⅈ * DE], 2/3]],
   Abs[Cos[(RR*ξ)/Sqrt[1+ⅈ*DE]]/Cos[RR/Sqrt[1+ⅈ*DE]]]}, {ξ, 0, 1}, PlotRange → {{0, 1}, {0, 15}}], -90 Degree]
```

Out[14]=



Other important variables...

Base-to-top dynamic amplification: exact, eq.(14), and high-frequency asymptotics, eq.(15), $\left(r >> \left| \left(\frac{1-n}{2-n}\right)^2 - \frac{1}{4} \right|\right)$

In[15]:=
$$A[r\_, n\_] = \frac{1}{\text{BesselJ}\left[-\frac{1-n}{2-n}, \frac{2\,r}{2-n}\right]} * \frac{1}{\left(\frac{r}{2-n}\right)^{\frac{1-n}{2-n}} * \text{Gamma}\left[\frac{1}{2-n}\right]};$$

In[16]:=
$$\text{Ahf}[r\_, n\_] = \sqrt{\pi} * \frac{\left(\frac{r}{2-n}\right)^{\frac{n}{2\,(2-n)}}}{\text{Gamma}\left[\frac{1}{2-n}\right]} * \text{Sec}\left[\frac{2\,r}{2-n} - \left(\frac{n}{2-n}\right)\frac{\pi}{4}\right];$$

In[17]:=
$$\text{Env}[r\_, n\_, \delta\_] = 2\,\frac{\text{Abs}\left[\sqrt{\pi} * \frac{\left(\frac{r}{2-n}\right)^{\frac{n}{2\,(2-n)}}}{\text{Gamma}\left[\frac{1}{2-n}\right]}\right]}{\text{Cosh}\left[\frac{2\,r}{2-n} * \delta\right]};$$

Condition for validity of the asymptotic approximation (Figure 3 in the text):

In[18]:=
```
figCon2 = Labeled[
  Plot[(2 - n)/2 Abs[((1 - n)/(2 - n))^2 - 1/4], {n, 0, 2}, PlotRange → {{0, 2}, {0, 1}}, RotateLabel → False,
   LabelStyle → (FontSize → 12), Ticks → {{0, 0.5, 1, 1.5, 2}, Automatic}],
  {"eq.(16) rhs", "n"}, {Left, Bottom}, LabelStyle → {Italic, FontSize → 18}]
```

Out[18]=



The following plot provides a quick verification:

In[19]:=
```
Plot[{Abs[u[0.000000001, r/Sqrt[1 + i * DE], 4/3]],
  Abs[A[r/Sqrt[1 + i * DE], 4/3]], Abs[Ahf[r/Sqrt[1 + i * DE], 4/3]]}, {r, 0, 10}]
```

Out[19]=



Main results:

Baseline Envelope: the following is a simple verification of eq.(17):

In[21]:=
```
TrigExpand[Refine[Expand[Sec[x - i * y]], {x ∈ Reals, y ∈ Reals}]]
```

Out[21]=

$$\frac{1}{\text{Cos}[x]\,\text{Cosh}[y] + i\,\text{Sin}[x]\,\text{Sinh}[y]}$$

And the following expression implements eq.(18)

In[22]:= $$\texttt{Env[r\_, n\_, }\delta\texttt{\_] = Sqrt[2]}\ \frac{\texttt{Abs}\left[\sqrt{\pi} * \dfrac{\left(\frac{r}{2-n}\right)^{\frac{n}{2\,(2-n)}}}{\texttt{Gamma}\left[\frac{1}{2-n}\right]}\right]}{\texttt{Cosh}\left[\frac{2\,r}{2-n} * \delta\right]^{1/2}};$$

A test plot:

In[23]:=
```
Ntest = 1.4;
DE = 0.3;
Plot[{Abs[A[r/Sqrt[1 + ⅈ * DE], Ntest]], Abs[Ahf[r/Sqrt[1 + ⅈ * DE], Ntest]], Env[r, Ntest, DE]},
 {r, 0, 10}, PlotRange → All, PlotLegends → "Expressions",
 Ticks → {{π/2 * (1 - Ntest / 4)}, Automatic}, GridLines → {{π/2 * (1 - Ntest / 4)}, None}]
```

Out[23]=


Exact solution with non-zero stiffness at the surface (Rovithis et al. 2011)

This result is included for verification purposes, as these authors showed that you one get the similar behavior using it with very large value of stiffness ratio between the base and the surface of the deposit.

These are the parameters that appear in the expresion:

In[24]:= $$\lambda\texttt{[}\alpha\texttt{\_, n\_, r\_, }\delta\texttt{\_] = }\frac{\texttt{r / Sqrt[}(1 + ⅈ * \delta)\texttt{]}}{(1 - n) * \left(1 - \alpha^{\frac{1}{2*n}}\right)};$$

In[25]:= $$\psi\texttt{[n\_] = }\frac{(1 - 2 * n)}{2};$$

In[26]:= $$\nu\texttt{[n\_] = }\frac{(2 * n - 1)}{2\,(1 - n)};$$

And this is the exact transfer function in the presence of finite stiffness at the surface:

In[27]:=
$$\text{TFR}[\alpha\_, n\_, r\_, \delta\_] = \frac{2}{\pi} * \frac{\left(\alpha^{\frac{1}{2*n}}\right)^{\psi[n]-1+n}}{\lambda[\alpha, n, r, \delta]} * \left(\text{BesselJ}\left[\nu[n]+1, \lambda[\alpha, n, r, \delta] * \left(\alpha^{\frac{1}{2*n}}\right)^{(1-n)}\right] * \text{BesselY}[\nu[n], \lambda[\alpha, n, r, \delta]] - \text{BesselY}\left[\nu[n]+1, \lambda[\alpha, n, r, \delta] * \left(\alpha^{\frac{1}{2*n}}\right)^{(1-n)}\right] * \text{BesselJ}[\nu[n], \lambda[\alpha, n, r, \delta]]\right)^{-1};$$

A quick verification plot: compare to homogeneous stratum

In[28]:=
```
DE = 0.1;
NN = 0.1;
Plot[{Abs[TFR[0.99, NN/2, r, DE]], Abs[Sec[r/Sqrt[1 + ⅈ * DE]]]},
 {r, 0, 5 * π/2}, PlotRange → {Automatic, {0, 14}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[28]=



Figure #4

First tile: n=1/3, δ=0.1

In[29]:=
```
DE = 0.1;
NN = 1/3;
Plot[{Abs[TFR[10^-6, NN/2, r, DE]], Abs[A[r/Sqrt[1 + i * DE], NN]], Env[r, NN, DE]},
 {r, 0, 5 * π/2}, PlotRange → {Automatic, {0, 14}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[29]=



Second tile: n=5/4, δ=0.1

In[31]:=
```
DE = 0.1;
NN = 5 / 4;
Plot[{Abs[TFR[10^-6, NN/2, r, DE]], Abs[A[r/Sqrt[1 + i * DE], NN]], Env[r, NN, DE]},
 {r, 0, 5 * π/2}, PlotRange → {Automatic, {0, 20}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[31]=



Third tile: n=1, δ=0.1

```
In[32]:= DE = 0.25;
NN = 15/9;
Plot[{Abs[TFR[10^-16, NN/2, r, DE]], Abs[A[r/Sqrt[1 + i * DE], NN]], Env[r, NN, DE]},
 {r, 0, 5 * π/2}, PlotRange → {Automatic, {0, 60}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[32]=



Figure #5

First tile: n=0.4, δ=0.3

```
In[33]:= DE = 0.3;
NN = 0.4;
Plot[{Abs[A[r/Sqrt[1 + i * DE], NN]], Abs[Ahf[r/Sqrt[1 + i * DE], NN]], Env[r, NN, DE]},
 {r, 0, 8 * π/2}, PlotRange → {Automatic, {0, 5}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[33]=



Second tile: n=1.5, δ=0.05

In[35]:=
```
DE = 0.05;
NN = 1.5;
Plot[{Abs[A[r/Sqrt[1 + ⅈ * DE], NN]], Abs[Ahf[r/Sqrt[1 + ⅈ * DE], NN]], Env[r, NN, DE]},
 {r, 0, 8 * π/2}, PlotRange → {Automatic, {0, 150}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[35]=



Third tile: n=1.75, δ=0.2

In[39]:=
```
DE = 0.2;
NN = 1.75;
Plot[{Abs[A[r/Sqrt[1 + ⅈ * DE], NN]], Abs[Ahf[r/Sqrt[1 + ⅈ * DE], NN]], Env[r, NN, DE]},
 {r, 0, 8 * π/2}, PlotRange → {Automatic, {0, 500}},
 PlotLegends → "Expressions", PlotStyle → {Automatic, {Orange, Dashed}}]
```

Out[39]=